Fortran IV program to calculate absorption and transmission of thermal radiation by single and double-glazed windows

Mitalas, G. P.; Arseneault, J. G.

Publisher's version / Version de l'éditeur:
https://doi.org/10.4224/20331063
DBR Computer Program, 1972-03-01

National Research Council Canada
Conseil national de recherches Canada

DIVISION OF BUILDING RESEARCH

FORTRAN IV PROGRAM TO CALCULATE ABSORPTION AND TRANSMISSION OF THERMAL RADIATION BY SINGLE AND DOUBLE - GLAZED WINDOWS

BY

G. P. MITALAS AND J. G. ARSENEAULT

DBR COMPUTER PROGRAM NO. 34

OTTAWA

MARCH 1972

CP 34

NATIONAL RESEARCH COUNCIL OF CANADA
DIVISION OF BUILDING RESEARCH

FORTRAN IV PROGRAM TO CALCULATE ABSORPTION AND TRANSMISSION OF THERMAL RADIATION BY SINGLE AND DOUBLE-GLAZED WINDOWS

by

G.P. Mitalas and J.G. Arseneault

Computer Program No. 34
of the
Division of Building Research

OTTAWA
March 1972

Fortran IV Program to Calculate Absorption and Transmission of Thermal Radiation by Single and Double-glazed Windows

G. P. Mitalas and J. G. Arseneault[1]

Division of Building Research
National Research Council of Canada
Ottawa

In the calculations of the heating or cooling load for a room, it is necessary to know the fraction of the solar radiation incident on the outside of the window that is absorbed by the glass and the fraction that is transmitted to the interior of the room. This program calculates the absorptivity and transmissivity of windows made of common glass. In addition the coefficients for a 5th degree polynomial are calculated to allow rapid (although less accurate) determination of these factors for a given window and given incident angle of solar beam.

The calculations by this program are based on:

(a) Fresnel's formulae (relation between incident angle, refraction angle and reflection of parallel and perpendicular polarization components of radiation).

(b) Snell's law (relation between refraction and incident angles).

(c) Exponential extinction law (relation between glass sheet thickness, extinction coefficient for glass and absorption of radiation in a single pass).

The calculated factors for single and double glazed windows account for the multiple reflections and absorptions that occur when radiation passes through more than one air-glass interface.

The 5th order polynomials, that relate the factors and the cosine of incident angle, are fitted by the least-squares method using the calculated values for incident angle 0° to 90° in one degree steps. The coefficients of these polynomials are used to calculate diffuse radiation factors assuming that the diffuse radiation from the sky and the ground has equal intensity at all incident angles.

Key Words: Absorbtion, glass, radiation, solar transmission, window.

---

[1] Research Officer and Computer Systems Programmer, respectively.

1. Expressions for Absorptivity and Transmissivity of a Window

The single air-glass interface reflectivity is given by Fresnel's formulae

$$r_{/\!/} = \frac{\tan^2(\theta_1 - \theta_2)}{\tan^2(\theta_1 + \theta_2)} \tag{1}$$

$$r_{\perp} = \frac{\sin^2(\theta_1 - \theta_2)}{\sin^2(\theta_1 + \theta_2)} \tag{2}$$

where $r_{/\!/}$ and $r_{\perp}$ = the reflectivity for radiation that is polarized with the electric vector parallel and perpendicular, respectively, to the plane that contains the incident beam and a normal to the interface.

θ_1 = incident angle

θ_2 = refraction angle

The refraction angle is related to the incident angle by Snell's law.

$$\text{Sin } \theta_1 = n \text{ Sin } \theta_2 \tag{3}$$

where n = index of refraction for the glass.

For normal incidence $\theta_1 = \theta_2 = 0$

and

$$r_{/\!/} = r_{\perp} = \left(\frac{n-1}{n+1}\right)^2 \tag{4}$$

The fraction of the radiation that is absorbed in a single pass through a glass sheet of thickness L is given by

$$a = 1 - \exp(-KL/\text{Cos } \theta_2) \tag{5}$$

where K = extinction coefficient for glass.

The absorbtivity, A, transmissivity, T, and reflectivity, R, of a sheet of glass and double-glazed window are calculated for parallel and perpendicular polarization separately. The average values of $A_{/\!/}$ and $A_{\perp}$, $T_{/\!/}$ and $T_{\perp}$, or $R_{/\!/}$ and $R_{\perp}$ are applicable for non-polarized incident beam.

The factors A, T, and R of a single sheet of glass (taking account of multiple reflections of both surfaces and multiple absorbtions) are given by

$$A = \frac{a(1-r)\,[1 + r(1-a)]}{1 - r^2(1-a)^2} \tag{6}$$

$$T = \frac{(1 - r)^2 (1 - a)}{1 - r^2 (1 - a)^2} \quad (7)$$

and

$$R = r + \frac{r(1 - r)^2 (1 - a)^2}{1 - r^2 (1 - a)^2} \quad (8)$$

where r and a are the single pass factors. The factors for parallel and perpendicular polarization components are calculated when $r = r_{//}$ and $r = r_{\perp}$ respectively.

The double-glazed window absorptivity, A_{1D} and A_{2D}, and transmissivity, T_D, are given by

$$A_{2D} = A_2 \left\{ 1 + R_1 T_2 \left(\frac{1}{1 - R_1 R_2} \right) \right\} \quad (9)$$

$$A_{1D} = \frac{A_1 T_2}{1 - R_1 R_2} \quad (10)$$

and

$$T_D = \frac{T_1 T_2}{1 - R_1 R_2} \quad (11)$$

where the quantities with subscript D are for double-glazed windows and those without subscript D are for a single sheet of glass. The subscript 2 denotes the factors for the outer pane and subscript 1 refers to the inner one.

2. Polynomial Coefficients

The 5th order polynomials, that relate the factors and the cosine of incident angle, are fitted by the least-squares method. For example, the polynomial that relates A and $\cos \theta_1$ is

$$A = \sum_{i=o}^{5} C_{A,i} (\cos \theta_1)^i \quad (12)$$

where $C_{A,i}$ = calculated polynomial coefficients for an absorbtivity A.

Polynomials are fitted only to the non-polarized beam factors using the calculated values for $\theta_1 = 0°$ to $90°$ in one degree steps.

3. Diffuse Radiation

For heat-gain calculations through a window it is usually assumed that the diffuse radiation from the sky or the ground has equal intensity at all incident angles. The factors for diffuse radiation of this nature are given by

$$F_{diffuse} = \int_o^{\pi/2} F(\theta) \sin 2\theta \, d\theta \quad (13)$$

where $F(\theta)$ is the factor for direct solar beam and is a function of the incident angle θ. The substitution of the polynomial expression for $F(\theta)$ and integration gives

$$F_{diffuse} = 2 \sum_{i=o}^{5} \frac{Ci}{i+2} \tag{14}$$

where Ci = 5th order polynomial coefficients that relates the factor F and $\cos \theta_1$.

4. General Description of the Program

This Fortran IV program is for an IBM-System/360 operating system.

The coding sheets, a sample of output and the flow diagram (fig. A1) are given on pages A-1 to A-7 of Appendix A.

Input:

Card 1 - columns 1-10 n, index of refraction

Card 2 - columns 1-10 KL inside
11-20 KL outside.

Format: Floating point, 10 columns.

This paper is a contribution of the Division of Building Research, National Research Council of Canada, and is published with the approval of the Director of the Division.

APPENDIX A

KL INSIDE=0.12   KL OUTSIDE=0.60   N=1.52

INCIDENT ANGLE	COSINE ANGLE	PERPENDICULAR A1	PERPENDICULAR A2	PERPENDICULAR T	PARALLEL A1	PARALLEL A2	PARALLEL T	AVERAGE A1	AVERAGE A2	AVERAGE T
0	1.0000	0.0569	0.4587	0.4114	0.0569	0.4587	0.4114	0.0569	0.4587	0.4114
10	0.9848	0.0568	0.4610	0.4067	0.0572	0.4606	0.4122	0.0570	0.4608	0.4095
20	0.9397	0.0565	0.4677	0.3922	0.0581	0.4661	0.4147	0.0573	0.4669	0.4034
30	0.8660	0.0558	0.4784	0.3667	0.0597	0.4751	0.4190	0.0578	0.4768	0.3929
40	0.7660	0.0542	0.4919	0.3284	0.0619	0.4871	0.4247	0.0581	0.4895	0.3765
45	0.7071	0.0529	0.4987	0.3036	0.0631	0.4940	0.4271	0.0580	0.4964	0.3653
50	0.6428	0.0511	0.5047	0.2746	0.0642	0.5015	0.4279	0.0577	0.5031	0.3513
55	0.5736	0.0486	0.5086	0.2411	0.0651	0.5095	0.4249	0.0569	0.5091	0.3330
60	0.5000	0.0454	0.5084	0.2033	0.0653	0.5184	0.4139	0.0553	0.5134	0.3086
65	0.4226	0.0410	0.5007	0.1617	0.0641	0.5281	0.3883	0.0526	0.5144	0.2750
70	0.3420	0.0355	0.4803	0.1178	0.0605	0.5375	0.3394	0.0480	0.5089	0.2286
72	0.3090	0.0329	0.4668	0.1002	0.0581	0.5400	0.3115	0.0455	0.5034	0.2058
74	0.2756	0.0300	0.4494	0.0828	0.0551	0.5407	0.2784	0.0425	0.4951	0.1806
76	0.2419	0.0269	0.4270	0.0660	0.0513	0.5382	0.2404	0.0391	0.4826	0.1532
78	0.2079	0.0234	0.3987	0.0502	0.0467	0.5303	0.1983	0.0351	0.4645	0.1243
80	0.1736	0.0197	0.3631	0.0357	0.0412	0.5136	0.1537	0.0305	0.4384	0.0947
82	0.1392	0.0156	0.3187	0.0230	0.0348	0.4836	0.1088	0.0252	0.4011	0.0659
84	0.1045	0.0113	0.2633	0.0126	0.0272	0.4329	0.0666	0.0193	0.3481	0.0396
86	0.0698	0.0068	0.1943	0.0051	0.0184	0.3506	0.0310	0.0126	0.2725	0.0181
88	0.0349	0.0025	0.1083	0.0009	0.0082	0.2179	0.0071	0.0054	0.1631	0.0040

POLYNOMIAL COEFFICIENTS FOR AVERAGE ABSORPTION, TRANSMISSION, AND REFLECTION.

	A1	A2	T	R
C0	-0.002607	0.035383	-0.006516	0.973741
C1	0.241959	4.179491	0.137991	-4.559441
C2	-0.316419	-14.234123	4.068337	10.482203
C3	0.091218	23.624634	-10.116615	-13.599236
C4	0.107813	-19.170715	9.522376	9.540527
C5	-0.065152	6.026407	-3.195348	-2.765906
DIFFUSE VALUES	0.054299	0.486093	0.334166	0.125438

LEGEND
A1=ABSORPTION INSIDE PANE
A2=ABSORPTION OUTSIDE PANE
T=TRANSMISSION
R=REFLECTION
IF KL OUTSIDE=0.00 THEN THE VALUES ARE FOR A SINGLE GLAZED WINDOW
KL=PRODUCT OF GLASS ABSORPTION COEFFICIENT AND GLASS THICKNESS
N=INDEX OF REFRACTION OF THE GLASS



Figure A1 - Flow Diagram for Absorption and Transmission Factor Calculation Program.

```
C     PROGRAM BR-81.
C     ABSORPTION AND TRANSMISSION OF THERMAL RADIATION
C     BY SINGLE AND DOUBLE GLAZED WINDOWS.
      DIMENSION A(100),B(100),C(100),D(100),E(100),F(100),X(100),
     9Y(100),Z(100),G(100),STORE(16,15),W(100),CC(4,7)
      DOUBLE PRECISION PI,X,Y,Z,G,STORE,W,TOL
      INTEGER CARD,PRINT
      REAL N,KLI,KLO
      CARD=1
      IOUT=2
      PRINT=3
      PI=3.14159265
      TOL=0.0000001
 2000 READ(CARD,1) N
    1 FORMAT(2F10.4)
      IF(N)10,1000,10
   10 READ(CARD,1)KLI,KLO
      WRITE(PRINT,5) KLI,KLO,N
    5 FORMAT(1H1,'KL INSIDE=',F4.2,4X,'KL OUTSIDE=',F4.2,4X,'N=',F4.2)
      WRITE(PRINT,1)
      DO 20 I=1,90
      THETA=(I-1)*PI/180.0
      G(I)=COS(THETA)
      IF(THETA)30,40,30
   40 RPA=(N-1.0)/(N+1.0)
      RPA=RPA*RPA
      RPE=RPA
      GO TO 50
   30 TETA=SIN(THETA)/N
      TEMP=SQRT(1.0-TETA*TETA)
      TETA=ATAN(TETA/TEMP)
      RPE=SIN(THETA-TETA)/SIN(THETA+TETA)
      RPA=RPE*COS(THETA+TETA)/COS(THETA-TETA)
      RPA=RPA*RPA
      RPE=RPE*RPE
   50 TEMP=1.0-EXP(-N*KLI/SQRT(N*N-SIN(THETA)*SIN(THETA)))
      APA=TEMP
      APE=TEMP
      IF(KLO)60,70,60
   60 TEMP=1.0 -EXP(-N*KLO/SQRT(N*N-SIN(THETA)*SIN(THETA)))
      APAO=TEMP
      APEO=TEMP
   70 TEMP=1.0 -RPA*RPA*(1.0-APA)*(1.0-APA)
      AIPA=APA*(1.0-RPA)*(1.0+RPA*(1.0-APA))/TEMP
      TIPA=(1.0-RPA)*(1.0-RPA)*(1.0-APA)/TEMP
      TEMP=1.0-RPE*RPE*(1.0-APE)*(1.0-APE)
      AIPE=APE*(1.0-RPE)*(1.0+RPE*(1.0-APE))/TEMP
      TIPE=(1.0-RPE)*(1.0-RPE)*(1.0-APE)/TEMP
      RIPE=1.0-AIPE-TIPE
      RIPA=1.0-AIPA-TIPA
      TI=(TIPE+TIPA)/2.0
      AI=(AIPE+AIPA)/2.0
      A(I)=AIPE
```

```
      B(I)=0.0
       C(I)=TIPE
      D(I)=AIPA
      E(I)=0.0
      F(I)=TIPA
      X(I)=AI
      Y(I)=0.0
      Z(I)=TI
      IF(KLO)80,20,80
   80 TEMP=1.0-RPA*RPA*(1.0-APAO)*(1.0-APAO)
      AOPA=APAO*(1.0-RPA)*(1.0+RPA*(1.0-APAO))/TEMP
      TOPA=(1.0-RPA)*(1.0-RPA)*(1.0-APAO)/TEMP
      TEMP=1.0-RPE*RPE*(1.0-APEO)*(1.0-APEO)
      AOPE=APEO*(1.0-RPE)*(1.0+RPE*(1.0-APEO))/TEMP
      TOPE=(1.0-RPE)*(1.0-RPE)*(1.0-APEO)/TEMP
      ROPE=1.0-AOPE-TOPE
      ROPA=1.0-AOPA-TOPA
      AO=AOPE*(1.0+RIPE*TOPE/(1.0-RIPE*ROPE))
      B(I)=AO
      AO=AO+AOPA*(1.0+RIPA*TOPA/(1.0-RIPA*ROPA))
      E(I)=AO-B(I)
      Y(I)=AO/2.0
      AI=AIPE*TOPE/(1.0-ROPE*RIPE)
      A(I)=AI
      AI=AI+AIPA*TOPA/(1.0-ROPA*RIPA)
      D(I)=AI-A(I)
      X(I)=AI/2.0
      TI=TOPE*TIPE/(1.0-RIPE*ROPE)
      C(I)=TI
      TI=TI+TOPA*TIPA/(1.0-RIPA*ROPA)
      F(I)=TI-C(I)
      Z(I)=TI/2.0
   20 CONTINUE
      CALL POLY(G,X,W,STORE,90,2,2,2,TOL,5)
      DO 90 I=1,6
      CC(2,I)=0.0
   90 CC(1,I)=STORE(I,5)
      CALL POLY(G,Z,W,STORE,90,2,2,2,TOL,5)
      DO 100 I=1,6
  100 CC(3,I)=STORE(I,5)
      IF(KLO)110,120,110
  110 CALL POLY(G,Y,W,STORE,90,2,2,2,TOL,5)
      DO 130 I=1,6
  130 CC(2,I)=STORE(I,5)
  120 WRITE(PRINT,2)
    2 FORMAT(1X,'INCIDENT',2X,'COSINE',7X,'PEPPENDICULAR',16X,
     9'PARALLEL',19X'AVERAGE')
      WRITE(PRINT,3)
    3 FORMAT(2X,'ANGLE',5X,'ANGLE',5X,'A1',6X,'A2',7X,'T',
     98X,'A1',6X,'A2',7X,'T',8X,'A1',6X,'A2',7X,'T')
      WRITE(PRINT,1)
      DO 140 I=1,41,10
      IT=I-1
```

```
0101         140 WRITE(PRINT,4) IT,G(I),A(I),B(I),C(I),D(I),E(I),F(I),X(I),
                9Y(I),Z(I)
0102             WRITE(PRINT,1)
0103           4 FORMAT(1X,I5,F11.4,F9.4,2F8.4,F10.4,2F8.4,F10.4,2F8.4)
0104             DO 150 I=46,66,5
0105             IT=I-1
0106         150 WRITE(PRINT,4) IT,G(I),A(I),B(I),C(I),D(I),E(I),F(I),X(I),
                9Y(I),Z(I)
0107             WRITE(PRINT,1)
0108             DO 160 I=71,79,2
0109             IT=I-1
0110         160 WRITE(PRINT,4) IT,G(I),A(I),B(I),C(I),D(I),E(I),F(I),X(I),
                9Y(I),Z(I)
0111             WRITE(PRINT,1)
0112             DO 165 I=81,89,2
0113             IT=I-1
0114         165 WRITE(PRINT,4) IT,G(I),A(I),B(I),C(I),D(I),E(I),F(I),X(I),
                9Y(I),Z(I)
0115             DO 170 I=1,100
0116         170 X(I)=1.0 -X(I)-Y(I)-Z(I)
0117             CALL POLY(G,X,W,STORE,90,2,2,2,TOL,5)
0118             DO 180 I=1,6
0119         180 CC(4,I)=STORE(I,5)
0120             DO 190 I=1,4
0121         190 CC(I,7)=0.0
0122             DO 200 I=1,6
0123             II=I+1
0124             DO 200 J=1,4
0125         200 CC(J,7)=CC(J,7)+CC(J,I)/II*2.0
0126             WRITE(PRINT,6)
0127           6 FORMAT(1H0,'POLYNOMIAL COEFFICIENTS FOR AVERAGE ABSORPTION, TRANSM
                9ISSION, AND REFLECTION.')
0128             WRITE(PRINT,1)
0129             WRITE(PRINT,8)
0130           8 FORMAT(25X,'A1',9X,'A2',10X,'T',10X,'R')
0131             WRITE(PRINT,7) ((CC(I,J),I=1,4),J=1,7)
0132           7 FORMAT(6X,'C0',11X,4F11.6/6X,'C1',11X,4F11.6/6X,'C2',11X,4F11.6/
                96X,'C3',11X,4F11.6/6X,'C4',11X,4F11.6/6X,'C5',11X,4F11.6/
                9            /1X,'DIFFUSE VALUES',4X,4F11.6)
0133             WRITE(IOUT,11) ((CC(I,J),J=1,7),I=1,4)
0134          11 FORMAT(7F11.6)
0135             WRITE(PRINT,1)
0136             WRITE(PRINT,9)
0137           9 FORMAT(1X,'LEGEND'/1X,'A1=ABSORPTION INSIDE PANE'/
                91X,'A2=ABSORPTION OUTSIDE PANE'/
                91X,'T=TRANSMISSION'/
                91X,'R=REFLECTION'/
                91X,'IF KL OUTSIDE=0.00 THEN THE VALUES ARE FOR A SINGLE GLAZED WIN
                9DOW'/1X,'KL=PRODUCT OF GLASS ABSORPTION COEFFICIENT AND GLASS THIC
                9KNESS'/1X,'N=INDEX OF REFRACTION OF THE GLASS')
0138             GO TO 2000
0139        1000 CALL EXIT
0140             END
```

```
             C   MODIFICAT+ON TO GRAVES POLYNOMIAL CURVE FITTING TO FORTRAN  IV
             C      N IS THE NUMBER OF OBSERVATIONS                                  GA  0046
             C     TOL IS THE ALLOWABLE VALUE OF THE STANDARD ERROR                  GA  0048
             C   LAST IS THE VALUE OF THE HIGHEST ORDER OF THE POLYNOMIAL TO BE FITTEDGA 0050
             C   III=1 FOR WEIGHTED INPUT                                            GA  0052
             C    III=2 FOR UNWEIGHTED INPUT                                         GA  0054
             C   ISS2=1 FOR INTERMEDIATE OUTPUT                                      GA  0056
             C   ISS2=2 FOR FINAL OUTPUT ONLY                                        GA  0058
             C   ISS3=1 FOR OUTPUT OF OBSERVED AND CALCULATED VALUES                 GA  0060
             C   ISS3=2 FOR NO OUTPUT                                                GA  0062
             C     ISS3 ONLY USED IF ISS2=1                                          GA  0064
 0001               SUBROUTINE POLY(X,Y,W,STORE,N,III,ISS2,ISS3,TOL,LAST)            GA  0066
 0002               DOUBLE PRECISION X(100),Y(100),A(16,31),SUMX(31),SUMY(15),W(100),CGA 0068
                   1,S1,S2,S3,B,TOL,E(16),S4,STORE(16,15)                            GA  0070
 0003               IN=1                                                             GA  0072
 0004               IOUT=3                                                           GA  0074
 0005               DO 1 I=1,16                                                      GA  0076
 0006               DO 1 J=1,15                                                      GA  0078
 0007             1 STORE(I,J)=0.0                                                   GA  0080
 0008               GO TO (70,50) , III                                              GA  0082
 0009            50 DO 60 I=1,N                                                      GA  0084
 0010            60 W(I)=1.                                                          GA  0086
 0011            70 SUMX(1)=0.                                                       GA  0088
 0012               SUMX(2)=0.                                                       GA  0090
 0013               SUMX(3)=0.                                                       GA  0092
 0014               SUMY(1)=0.                                                       GA  0094
 0015               SUMY(2)=0.                                                       GA  0096
 0016               DO 90 I=1,N                                                      GA  0098
 0017               SUMX(1)=SUMX(1)+W(I)                                             GA  0100
 0018               SUMX(2)=SUMX(2)+W(I)*X(I)                                        GA  0102
 0019               SUMX(3)=SUMX(3)+W(I)*X(I)*X(I)                                   GA  0104
 0020               SUMY(1)=SUMY(1)+W(I)*Y(I)                                        GA  0106
 0021            90 SUMY(2)=SUMY(2)+W(I)*X(I)*Y(I)                                   GA  0108
 0022               NORD=1                                                           GA  0110
 0023            91 L=NORD+1                                                         GA  0112
 0024               KK=2*L+1                                                         GA  0114
 0025               KA=L+1                                                           GA  0116
 0026               DO 101 I=1,L                                                     GA  0118
 0027               DO 100 J=1,L                                                     GA  0120
 0028               IK=J-1+I                                                         GA  0122
 0029               JL=J+L+1                                                         GA  0124
 0030               A(I,JL)=0.                                                       GA  0126
 0031       100     A(I,J)=SUMX(IK)                                                  GA  0128
 0032               JL=I+L+1                                                         GA  0130
 0033               A(I,JL)=1.0                                                      GA  0132
 0034       101     A(I,KA)=SUMY(I)                                                  GA  0134
 0035               DO 140 I=1,L                                                     GA  0136
 0036               A(KA,I)=-1                                                       GA  0138
 0037               KKK=I+1                                                          GA  0140
 0038               DO 110 J=KKK,KK                                                  GA  0142
 0039       110     A(KA,J)=0.                                                       GA  0144
 0040               C=1./A(1,I)                                                      GA  0146
 0041               DO 120 II=2,KA                                                   GA  0148
                                                                                     GA  0150
```

```
          DO 120 J=KKK,KK
    120   A(II,J)=A(II,J)-A(1,J)*A(II,I)*C
          DO 140 II=1,L
          DO 140 J=KKK,KK
    140   A(II,J)=A(II+1,J)
          DO 141 II=1,L
          J=NORD
      141 STORE(II,J)=A(II,KA)
C TO CALCULATE THE ERROR IN THE COEFFICIENTS YOU NOW NEED
          S4=0.
          DO 160 J=1,N
          S1=A(1,KA)
          DO 150 I=1,NORD
    150   S1=S1+A(I+1,KA)*X(J)**I
      160 S4=S4+W(J)*(S1-Y(J))*(S1-Y(J))
          B=N-L
          S4=(S4/B)**.5
          S2=S4
          DO 1000 I=1,L
          J=I+L
    1000  E(I)=(A(I,J+1   ))**.5*S4
          GO TO (163,161),ISS2
      161 IF(NORD-LAST)162,173,162
      162 IF(S2-TOL)163,163,171
      163 WRITE (IOUT,10)
       10 FORMAT (1H0,5HORDER,5X9HTOLERANCE,15X14HSTANDARD ERROR,6X3HOBS)
          WRITE (IOUT,8) NORD,TOL,S2,N
        8 FORMAT (1X  I3,2D24.16,I3//)
      210 DO 164 I=1,L
          J=I-1
          IF (I-1) 164,11,164
       11 WRITE (IOUT,12)
       12 FORMAT (8X12HCOEFFICIENTS,14X10HSTD. ERROR,/)
      164 WRITE (IOUT,6) J,A(I,KA),E(I)
        6 FORMAT (1XI3,D24.16,D24.16)
          GO TO (167,165),ISS3
      165 IF(NORD-LAST)166,167,166
      166 IF(S2-TOL)167,167,171
      167 DO 169 I=1,N
          S1=A(1,KA)
          DO 168 J=1,NORD
      168 S1=S1+A(J+1,KA)*X(I)**J
          S3=Y(I)-S1
          IF (I-1) 169,14,169
       14 WRITE(IOUT,15)
       15 FORMAT (1H0,9X6HOBS(X),17X6HOBS(Y),17X6HCAL(Y),17X10HDIFFERENCE,/)
      169 WRITE(IOUT,7) X(I),Y(I),S1,S3
        7 FORMAT (1X4D24.16)
          IF(NORD-LAST)170,173,173
      170 IF(S2-TOL)173,173,171
      171 NORD=NORD+1
          J=2*NORD
          SUMX(J)=0.
```

```
      SUMX(J+1)=0.                                                   GA  0258
      SUMY(NORD+1)=0.                                                GA  0260
      DO 172 I=1,N                                                   GA  0262
      SUMX(J)=SUMX(J)+X(I)**(J-1)*W(I)                               GA  0264
      SUMX(J+1)=SUMX(J+1)+X(I)**J*W(I)                               GA  0266
  172 SUMY(NORD+1)=SUMY(NORD+1)+Y(I)*X(I)**NORD*W(I)                 GA  0268
      GO TO 91                                                       GA  0270
  173 CONTINUE                                                       GA  0272
  200 RETURN                                                         GA  0274
      END                                                            GA  0276
```